Calculadora SAP – OnLine
Rodrigo Gama

Veremos como criar uma calculadora no SAP semelhante a do Windows utilizando Online. É um bom exercício para aprender online. Para maiores informações sobre o funcionamento, pesquisar sobre a calculadora do Windows. 

Este é um código que pode ser aprimorado e não esta 100% livre de erros.

Pela SE80, criar o programa abaixo.:

Após a criação, colar o código abaixo, salvar e ativar:

*&---------------------------------------------------------------------*
*& Report  Z_RODRIGO_TESTE3
*&---------------------------------------------------------------------*
REPORT  z_rodrigo_teste3.

DATA: vg_okcode(4)   TYPE c,
      wc_display(38) TYPE c,
      wc_dismem(5)   TYPE c,
      wc_memoria(38) TYPE c,
      wc_oper(1)     TYPE c,
      wc_aux(38)     TYPE c,
      wc_aux2(38)    TYPE c,
      wc_aux3(2)     TYPE c.

*&---------------------------------------------------------------------*
*&      Module  STATUS_0100  OUTPUT
*&---------------------------------------------------------------------*
MODULE status_0100 OUTPUT.
  SET PF-STATUS 'STATUS_0'.
  SET TITLEBAR '001'.
ENDMODULE. "STATUS_0100  OUTPUT

*&---------------------------------------------------------------------*
*&      Module  USER_COMMAND_0100  INPUT
*&---------------------------------------------------------------------*
MODULE user_command_0100 INPUT.

  CASE vg_okcode.
*&---------------------------------------------------------------------*
* Comandos do programa, sair, voltar, etc.
*&---------------------------------------------------------------------*
    WHEN 'EXIT' OR 'BACK' OR 'UP'.
      CLEAR vg_okcode.
      LEAVE PROGRAM.

*&---------------------------------------------------------------------*
* Botões Numéricos
*&---------------------------------------------------------------------*
    WHEN 'BT0'.
      IF wc_display NE 0.
        CONCATENATE wc_display '0' INTO wc_display.
      ENDIF.
    WHEN 'BT1'.
      CONCATENATE wc_display '1' INTO wc_display.
    WHEN 'BT2'.
      CONCATENATE wc_display '2' INTO wc_display.
    WHEN 'BT3'.
      CONCATENATE wc_display '3' INTO wc_display.
    WHEN 'BT4'.
      CONCATENATE wc_display '4' INTO wc_display.
    WHEN 'BT5'.
      CONCATENATE wc_display '5' INTO wc_display.
    WHEN 'BT6'.
      CONCATENATE wc_display '6' INTO wc_display.
    WHEN 'BT7'.
      CONCATENATE wc_display '7' INTO wc_display.
    WHEN 'BT8'.
      CONCATENATE wc_display '8' INTO wc_display.
    WHEN 'BT9'.
      CONCATENATE wc_display '9' INTO wc_display.

*&---------------------------------------------------------------------*
* Botões de Operação
*&---------------------------------------------------------------------*
* Botão Somar
    WHEN 'BTSM'.
      wc_aux = wc_display.
      wc_oper = '+'.
      CLEAR wc_display.

* Botão Subtrair
    WHEN 'BTSB'.
      wc_aux = wc_display.
      wc_oper = '-'.
      CLEAR wc_display.

* Botão Multiplicar
    WHEN 'BTMT'.
      wc_aux = wc_display.
      wc_oper = '*'.
      CLEAR wc_display.

* Botão Dividir
    WHEN 'BTDV'.
      wc_aux = wc_display.
      wc_oper = '/'.
      CLEAR wc_display.

* Botão Raiz Quadrada
    WHEN 'BTSQ'.
      IF wc_display IS NOT INITIAL.
        TRY.
            wc_display = SQRT( wc_display ).
          CATCH cx_sy_conversion_no_number.
            wc_display = 'ERROR'.
          CATCH cx_sy_arithmetic_error.
            wc_display = 'ERROR'.
          CATCH cx_sy_conversion_error.
            wc_display = 'ERROR'.
        ENDTRY.
      ENDIF.

* Botão Porcentagem
    WHEN 'BTPR'.
      IF wc_oper = '*' AND wc_display IS NOT INITIAL.
        wc_aux2 = ( wc_aux * wc_display ) / 100.
        wc_display = wc_aux2.
        CLEAR: wc_aux, wc_aux2.
      ENDIF.

* Botão Divide por 1.
    WHEN 'BTD1'.
      IF wc_display IS NOT INITIAL.
        wc_display = 1 / wc_display.
      ENDIF.

*&---------------------------------------------------------------------*
* Botões de Funcionais
*&---------------------------------------------------------------------*
* Botão casa Decimal.
    WHEN 'BTPT'.
      IF wc_display IS INITIAL.
        CONCATENATE wc_display '0.' INTO wc_display.
      ENDIF.
* Só permite um ponto decimal.
      IF wc_display NA '.'.
        CONCATENATE wc_display '.' INTO wc_display.
      ENDIF.

* Botão Limpar
    WHEN 'BTC'.
      CLEAR: wc_display, wc_aux, wc_aux2, wc_oper.

* Botão Backspace.
    WHEN 'BTBK'.
      IF wc_display+1(1) IS INITIAL.
        CLEAR wc_display.
      ELSE.
        wc_aux3 = 37.
        DO 37 TIMES.
          IF wc_display+wc_aux3(1) IS NOT INITIAL.
            CLEAR wc_display+wc_aux3(1).
            CLEAR wc_aux3.
            EXIT.
          ENDIF.
          wc_aux3 = wc_aux3 - 1.
        ENDDO.
      ENDIF.

* Botão de alteração do sinal.
    WHEN 'BTSN'.
      IF wc_display IS NOT INITIAL AND wc_display NE 0.
        IF wc_display NA '-'.
          wc_aux = wc_display.
          CLEAR wc_display.
          wc_display = '-'.
          CONCATENATE wc_display wc_aux INTO wc_display.
          CLEAR wc_aux.
        ELSE.
          REPLACE '-' WITH '' INTO wc_display.
          CONDENSE wc_display NO-GAPS.
        ENDIF.
      ENDIF.

*&---------------------------------------------------------------------*
* Botões de memória
*&---------------------------------------------------------------------*
* Botão para adicionar valor a memória MS
    WHEN 'BTAD'.
      IF wc_display IS NOT INITIAL.
        wc_memoria = wc_display.
        wc_dismem = '  M  '.
      ENDIF.

* Botão para limpar valor da memória
    WHEN 'BTCL'.
      CLEAR: wc_memoria, wc_dismem.

* Botão para exibir valor memorizado.
    WHEN 'BTEX'.
      wc_display = wc_memoria.

* Botão para adicionar valor da tela ao valor memorizado
    WHEN 'BTSO'.
      TRY.
          wc_memoria = wc_memoria + wc_display.
        CATCH cx_sy_conversion_overflow.
          wc_display = 'ERROR'.
        CATCH cx_sy_arithmetic_overflow.
          wc_display = 'ERROR'.
        CATCH cx_sy_conversion_no_number.
          wc_display = 'ERROR'.
      ENDTRY.

*&---------------------------------------------------------------------*
* Botão Resultado '='.
*&---------------------------------------------------------------------*
    WHEN 'BTEQ'.
      CASE wc_oper.

* Operação de Soma
        WHEN '+'.
          wc_aux2 = wc_display.
          CLEAR wc_display.

          TRY.
              wc_display = wc_aux + wc_aux2.
            CATCH cx_sy_conversion_overflow.
              CLEAR: wc_aux, wc_aux2.
              wc_display = 'ERROR'.
            CATCH cx_sy_arithmetic_overflow.
              CLEAR: wc_aux, wc_aux2.
              wc_display = 'ERROR'.
            CATCH cx_sy_conversion_no_number.
              wc_display = 'ERROR'.
          ENDTRY.

          CLEAR: wc_aux, wc_aux2, wc_oper.

* Operação de subtração
        WHEN '-'.
          CONDENSE wc_aux.
          wc_aux2 = wc_display.
          CLEAR wc_display.

          TRY.
              wc_display = wc_aux - wc_aux2.
            CATCH cx_sy_conversion_overflow.
              CLEAR: wc_aux, wc_aux2.
              wc_display = 'ERROR'.
            CATCH cx_sy_arithmetic_overflow.
              CLEAR: wc_aux, wc_aux2.
              wc_display = 'ERROR'.
            CATCH cx_sy_conversion_no_number.
              wc_display = 'ERROR'.
          ENDTRY.

          IF wc_aux < wc_aux2 AND wc_aux IS NOT INITIAL AND wc_aux2 IS NOT INITIAL.
            CONCATENATE wc_display '-' INTO wc_display.
          ENDIF.
          CLEAR: wc_aux, wc_aux2, wc_oper.

* Operação de Multiplicação
        WHEN '*'.
          wc_aux2 = wc_display.
          CLEAR wc_display.

          TRY.
              wc_display = wc_aux * wc_aux2.
            CATCH cx_sy_conversion_overflow.
              CLEAR: wc_aux, wc_aux2.
              wc_display = 'ERROR'.
            CATCH cx_sy_arithmetic_overflow.
              CLEAR: wc_aux, wc_aux2.
              wc_display = 'ERROR'.
            CATCH cx_sy_conversion_no_number.
              wc_display = 'ERROR'.
          ENDTRY.

          CLEAR: wc_aux, wc_aux2, wc_oper.

* Operação de Divisão
        WHEN '/'.
          wc_aux2 = wc_display.
          CLEAR wc_display.

          TRY.
              wc_display = wc_aux / wc_aux2.
            CATCH cx_sy_conversion_overflow.
              CLEAR: wc_aux, wc_aux2.
              wc_display = 'ERROR'.
            CATCH cx_sy_arithmetic_overflow.
              CLEAR: wc_aux, wc_aux2.
              wc_display = 'ERROR'.
            CATCH cx_sy_conversion_no_number.
              wc_display = 'ERROR'.
          ENDTRY.

          CLEAR: wc_aux, wc_aux2, wc_oper.

      ENDCASE.
  ENDCASE.
ENDMODULE. "USER_COMMAND_0100  INPUT
De dois cliques no 'STATUS_0' da linha de código SET PF-STATUS 'STATUS_0'. Para definir os botões do topo da tela.
A mensagem abaixo deve ser exibida, clicar em sim.

Informar o Texto breve

Preencher os botões BACK, UP e EXIT – Salvar e ativar

De dois cliques no '001' da linha de código SET TITLEBAR '001'. Para definir o Titulo do relatório.

O Titulo pode ser Calculadora SAP.

Após isso devemos criar a tela da Calculadora, clicar com o botão direito sobre o objeto, Criar, Tela. Conforme a imagem abaixo:

Informar o número da tela, no caso 100.

Informar a descrição e salvar

Clicar na aba “LógProcmto.” e descomentar os dois módulos PAI e PBO que definem o processamento da tela. Salvar e ativar.

Podemos clicar duas vezes sobre MODULE STATUS_0100. e MODULE USER_COMMAND_0100. e veremos que já estão criados.
Agora, vamos a parte mais trabalhosa que é criar o layout da calculadora. Ainda no processamento da tela 100, clicar no botão “Layout” na parte superior da tela.

Criar um layout semelhante a este:

Clicar no campo texto e arrastar sobre a tela, ao fazermos isso um campo rosa é exibido, clicar duas vezes sobre este campo

Escrever no campo “Texto” Calculadora SAP e sair da caixa de dialogo.

O resultado deve ser este:

Obs.: Todos os campos e botões inseridos nesta tela são feitos desta forma.
Selecionar a ferramenta “Quadro” e reproduzir o quadro abaixo que será a moldura da calculadora.

Clicar na ferramenta Campo de Entrada / Saída. Arrastar até o tamanho desejado no caso esta com 38 posições.
Em atributos, desmarcar a opção “Campo de entrada”.

Informar o nome do campo, no caso WC_DISPLAY que irá informar os valores a serem calculados, a declaração deste campo deve estar no programa. A mesma coisa deve ser feita para o campo WC_DISMEM conforme imagem abaixo.

Criar o campo WC_DISMEM que informa se a memória da calculadora esta em uso. Em atributos, desmarcar a opção “Campo de entrada”.

Clicar na opção “Botão” e criar o botão Backspace conforme tela abaixo.

Obs.: Este processo de criar botão será o mesmo para todos os botões da calculadora, somente mudando o Nome, texto, Cód.função e InfoRápida (Ao passar o mouse sobre o botão, a informação inserida neste campo será exibida) 

Obs2.: O campo Cód.função [image: image31.png]Cédfungio [FTEK  CigFung




 é o que define o processamento. Ao apertar o botão backspace, o valor BTBK será utilizado na variável de controle VG_OKCODE (Será vista mais adiante) e dentro do case o processamento do botão será executado conforme imagem abaixo.

Seqüência de criação de botões:

Obs.: Botões do 0 ao 7 são idênticos somente mudando os campo Nome, texto e Cód.função onde substituímos de acordo com o número. Ex.: Botão 7, nome: WB_BOTAO7, Texto: 7 e Cód.função:  BT7.

Após o layout criado, salvar e ativar. Sair do editor de layout. Ao entrar na tela, aba Lista de ementos, deve ser exibida a lista de elementos da tela conforme abaixo.

A variável VG_OKCODE declarada no programa, deve ser declarada também na lista de elementos da tela.

Salvar e ativar a tela.
Devemos criar uma transação. Clicar com o botão direito no nome do objeto, criar, transação.

Informar o nome do programa, no caso Z_CALC e a tela, no caso 100. Marcar os checks do quadro Capacidade GUI. Salvar.

Verificar se esta tudo ativo.

Para executar, podemos ir pelo caminho abaixo: Botão direito sobre a transação, executar, direto.

Ou informar o nome da transação na linha de comando.

A calculadora esta pronta para ser utilizada.

